Do Repeat Loops in SPSS

Using a “DO REPEAT” loop in SPSS can often reduce the amount of code that must be written, and may help reduce the likelihood of error. It is possible to create or modify many variables with fewer lines of code.

Following is an example of a do repeat. It uses twenty eight existing variables to create fourteen dichotomous variables. The variables in “RELATE” are existing variables that tell the relationship to head of fourteen people in the household. The variables in “STATUS” are existing variables that indicate whether each of the fourteen people live in the household, or only visit. The variables in “CH” are variables that are created by the SPSS statements. These variables will be coded 1 if the person lives in the household, and has the relationship code of “child of head”, or “step-child of head”. They will be coded as 0 if the person lives in the household, but is not a child or step-child of the head. They will be coded as missing if the person does not live in the household.

*************** !! Each variable list must have the same number of variables.
*************** !! The order of the variables is IMPORTANT.
*************** the variables P1Q3 P1Q4 child1 will be processed at the same time.
*************** the variables P2Q3 P2Q4 child2 will be processed at the same time.
do repeat
  Relate= P1Q3 P2Q3 P3Q3 P4Q3 P5Q3 P6Q3 P7Q3 P8Q3 P9Q3 P10Q3 P11Q3 P12Q3 P13Q3 P14Q3 /
  status= P1Q4 P2Q4 P3Q4 P4Q4 P5Q4 P6Q4 P7Q4 P8Q4 P9Q4 P10Q4 P11Q4 P12Q4 P13Q4 P14Q4 /
  ch = child1 to child14 .

  * Set the variables child1 to child14 to 0 .
  compute ch = 0 .

  *Set values in “ch” variables to 1 if corresponding code in “relate” variable is 3 or 4.
  if relate = 3 or relate = 4  ch = 1.

  * Set values of child1 to child14 to -9 if corresponding code in “status” variable is 2.
  if status = 2 ch = -9.
end repeat .

  * outside of do repeat loop, change values of -9 to missing values .
  missing values child1 to child14 (-9) .

This code could also be written as follows.

  * ALTERNATE WAY .
  * Set the variable child1 to 0 .
  compute child1 = 0 .

  *Set value of child1 1 if P1Q3 is 3 or 4.
  if P1Q3 = 3 or P1Q3 = 4  child1 = 1 .

  * Set value of child1 to -9 if P1Q4 is 2.
  if P1Q4 = 2 child1 = -9 .

  * Repeat this code after (ALTERNATE WAY) for of the fourteen people
  * in the household, changing variable names for each person.

  missing values child1 to child14 (-9) .

---

1Prepared by Patty Glynn, University of Washington. January 30, 2001, updated June 16, 2002 C:\all\help\helpnew\dorepeat.wpd